

Getting Started with Spice

By

Professor G. W. Neudeck

for


ECE 255

Fall 2002

1. Introduction.....	2
2. Example 1.....DC.....	3
3. Example 2..... I-measure.....	6
4. Example 3..... Diode circuit.....	8

1. Introduction

Spice is a very general circuit analysis program used extensively in industry and academia. The letters stand for Simulation Package for Integrated Circuit Emphasis (SPICE). There are many different versions available, some free and some with some costs as high as \$ 15,000. Each version has somewhat different capabilities, but we will need only the basic features, and basic device models, which are fairly standard. At Purdue, in the School of Electrical and Computer Engineering, has Spice-2G.6 installed on ECN directly or in the PUNCH simulation HUB. See the URL <http://nanohub.purdue.edu/> and sign up for an account. You will see this part of the screen.



Next click on the Spice2G and follow the instructions. Note there is a simple manual that YOU may print.

Inexpensive “student versions” for the Mac or PC are also available. The Bookstores carry several versions with manuals. Also the WWW as some for SPICE software for free. I use several different versions for different applications. Some have a screen entry for the circuit and good plotting capabilities.

The best way to learn to play tennis, soccer, basketball etc. is by doing it; the same is true with Spice. We expect you all to learn it as we go, however, most of you may have already used it. The following examples, along with purchasing a good Spice manual of your choice should get you started.

2. Example 1 DC circuit

Lets begin with a very simple linear circuit analysis with a dependent source. First there is a “file” of commands created with a text editor or on your Mac or PC a screen editor.

1. Header information
2. Circuit topology description
3. Commands to tell what type, frequencies, printing, plotting, etc. is to be performed
4. A command to tell it when to stop

Consider the dc circuit shown below. We wish to obtain the values of the dc voltage and dc current in R_1 and R_8 . The nodes are identified as squares with the node number inside. **The ground node must always be labeled node “0”**. The currents are given only through a voltage source in which Spice assumes that the current is positive when it flows into the positive node. Note that in the case of VDD, “in” means that the results should have a negative current listed for that source. (We know that the actual current must be coming out of VDD even though Spice says it is entering the positive node with a negative value.) We use a “dummy” voltage source of 0 value to measure a current, but first lets do an example without any dummy voltage sources.

Let the circuit of Figure 1 have the following component values. Note the voltage dependent current source has its value dependent on the voltage across R_4 .

$$R1 = 1K\Omega \quad R2 = 2K\Omega \quad R3 = 3.3k\Omega \quad R4 = 5K\Omega \quad R8 = 2.55K\Omega$$

$$VDD = 9 \text{ volts} \quad G = 0.001$$




Figure 1

The first step is to create a text file using your favorite editor or PC screen input topology. It should be noted that the comment line is signaled by an * and is ignored by Spice. To continue a command or component line to a second line, use a + sign (but not in a comment line). Do not leave a blank line at the end of the input file! Spell every word correctly, check the number of components on the diagram and lines in input file, and check all polarities.

The input file has the following form:

Example 1 for intro to Spice-2 DC

(1st line is a title)

*This is a comment line; below is the circuit description. **The () are NOT part of program!**

VDD 1 0 9 (positive node 1 to node 0; value 9 volts)
 R1 1 2 1000 (node 1 to 2; value 1K ohm)
 R2 2 3 2K (node 2 to 3; value 2K ohm)
 R3 4 0 3.3K (node 4 to 0; value 3.3K ohm)
 R4 2 0 5000 (node 2 to 0; value 5K ohm)
 R8 3 4 2.55K (node 3 to 4; value 2.55K ohm)
 GVR4 3 0 2 0 0.001 (node 0 to 3; value 1K ohm)

*The dependent current source nodes and polarity, and where the

* dependent voltage nodes are located and their polarity

*The convention for Spice is the current flows into the positive node on all sources

.OP (to tell Spice to do a DC analysis and print all node voltages and voltage source currents)

.END (to tell it to stop; no space after end)

The actual input text file looks like this:

Example 1 for intro to Spice-2 DC

*This is a comment line; below is the circuit description above is the title

VDD 1 0 9
 R1 1 2 1000
 R2 2 3 2K
 R3 4 0 3.3K
 R4 2 0 5000
 R8 3 4 2.55K
 GVR4 3 0 2 0 0.001

*The dependent current source nodes and polarity, and where the

* dependent voltage nodes are located and their polarity

*The convention for Spice is the current flows into the positive node on all sources

.OP

.END

The output file generally looks like this:

-----08/12/02 ----- SPICE 2g.6 -----20:45:20-----

Example 1 for intro to Spice-2 DC

----- Input Listing Temperature = 27.000 Deg C

*This is a comment line; below is the circuit description above is the title

VDD 1 0 9
R1 1 2 1000
R2 2 3 2K
R3 4 0 3.3K
R4 2 0 5000
R8 3 4 2.55K
GVR4 3 0 2 0 0.001

*The dependent current source nodes and polarity, and where the
* dependent voltage nodes are located and their polarity
*The convention for Spice is the current flows into the positive node on all sou
.OP
.END

-----08/12/02 ----- SPICE 2g.6 -----20:45:20-----

Example 1 for intro to Spice-2 DC

----- Small Signal Bias Solution Temperature = 27.000 Deg C

node voltage node voltage node voltage node voltage
(1) 9.0000 (2) 4.3423 (3) -3.2360 (4) -1.8254

Voltage source currents

Name current
vdd -4.658D-03

Total power dissipation 4.19D-02 Watts

-----08/12/02 ----- SPICE 2g.6 -----20:45:20-----

Example 1 for intro to Spice-2 DC

-----Operating Point Information Temperature = 27.000 Deg C

**** Voltage-Controlled Current Sources

gvr4
i-source 4.34E-03

Job concluded

Total job time 0.02

Now we have the information we need to obtain the current in R1 as 4.658 mA (by looking at $V_{DD} = -4.658$ mA into the source) and in R8 = $(\text{Node 4} - \text{Node 3})/2.55\text{K} = (-1.8254 - 3.2360)/2.55\text{K} = 0.55318$ mA. The voltage across R1 is $\text{Node 1} - \text{Node 2} = 9 - 4.3423 = 4.6577$ volts. Note that it checks with the current in R1. The voltage across R8 is $1.4106 = (\text{Node 4} - \text{Node 3})$.

3. Example 2 I-measure

Now consider the case of using a “dummy” voltage source to measure the current in R8. The circuit is modified, by adding a zero value voltage source, to ‘measure’ the current in R8.




Figure 2

The input file looks like this:

Example 2 for intro to Spice-2 Neudeck

```
VDD 1 0 9
R1 1 2 1000
R2 2 3 2K
R3 5 0 3.3K
R4 2 0 5000
R8 3 4 2.55K
GVR4 3 0 2 0 0.001
Vx 5 4 0
```

*The convention for Spice is the current flows into the positive node on all sources

```
.OP
.END
```

The output file looks like this:

-----08/13/02----- SPICE 2g.6 -----07:27:09-----

Example 2 for intro to Spice-2 Neudeck

----- Input Listing Temperature = 27.000 Deg C

```
VDD 1 0 9
R1 1 2 1000
R2 2 3 2K
R3 5 0 3.3K
R4 2 0 5000
R8 3 4 2.55K
GVR4 3 0 2 0 0.001
Vx 5 4 0
*The current flows into the positive node on all sources
.OP
.END
```

-----08/13/02----- SPICE 2g.6 -----07:27:09-----

Example 2 for intro to Spice-2 Neudeck

----- Small Signal Bias Solution Temperature = 27.000 Deg C

node voltage node voltage node voltage node voltage

(1) 9.0000 (2) 4.3423 (3) -3.2360 (4) -1.8254

(5) -1.8254

Voltage source currents

Name current

vdd -4.658D-03

vx 5.532D-04

Total power dissipation 4.19D-02 Watts

Example 2 for intro to Spice-2 Neudeck

----- Operating Point Information Temperature = 27.000 Deg C

**** Voltage-Controlled Current Sources

```
gvr4
i-source 4.34E-03
```

Job concluded

Total job time 0.02

4. Example 3 Diode

Consider the diode circuit shown. We wish to obtain the DC and small signal AC voltages across the diode and its ac and dc currents.




Figure 3.

The first step in a Spice analysis is to number the nodes in the circuit. Remember that ground **MUST** be **node 0**. With this information, the Spice input is written into a **text** file. For the circuit shown above, the Spice input file is:

```

Test Program for Diode Circuit in Problem Fig. 3
*This is a comment. In some SPICE outputs E=D for 10
.model diode1 d is=6e-9 n=1.9
d1 3 0 diode1
R1 2 3 500
V1 1 0 dc 6
Vs 2 1 ac 0.001
.dc V1 6 6.1 0.02
.tf v(3) V1
.ac lin 3 1.66 3.66
.OP
.print dc v(3,0) i(V1) v(1,0)
.print ac v(3,0) i(vs)
.END

```


Consult the Spice manual for detailed descriptions of the many more Spice commands. A brief description of the lines in this input file is as follows:

1. The first line **must** be the title.
2. Comments begin with a * in the first column.
3. The input & output width sets to 80 columns.(**.Width in=80 out=80**) **[line not used]**
4. The **model line** describes the parameters of a semiconductor device to Spice. The single **d** tells Spice that this is a diode model. Diode1 is the model name which can be anything you choose. Many more parameters than those shown are available for a more accurate non-linear model, small signal high frequency analysis, and transient analysis.
5. The diode **d1** has parameters of model diode1. Note the node polarity.
6. Resistor **r1** is between nodes 2 and 3 and has resistance 500 Ω .
7. **V1** is a 6 volt dc supply between nodes 1 and 0. {V1 1 0 dc 6}
8. **Vs** is a 1 mV ac supply between nodes 2 and 1. {Vs 2 1 ac 0.001}
9. **.dc V1 6 6.1 0.02** tells Spice to sweep the dc voltage of V1 from 6 to 6.1 in 0.01 volt steps. This is necessary for the **print dc** line below to work.
10. **.tf v(3) V1** does a incremental signal dc analysis with V1 as the input source and the voltage at node 3 as the output parameter. Note: Capacitors act as open circuits for this command, inductors are short circuits.
11. **.ac lin 3 1.66 3.66** sweeps the ac frequency from 1.66 to 3.66 Hz at 3 frequencies in a linear step. This is necessary so the “.print ac” line will work.
{ac lin 1 1.66 1.66 } can be used for one frequency.
12. **.OP** determines a dc operating point for the circuit.
13. **.print dc v(3,0) i(V1) v(1,0)** prints the dc voltage between nodes 3 and 0, the dc current from supply V1, and the dc voltage between nodes 1 and 0. The **.print dc** line is used with the .dc line. Note that Spice can only print the current for a supply.
14. **.print ac v(3,0) i(Vs)** The ac voltage between nodes 3 and 0 is printed; the ac current from supply Vs is also printed.
15. The last line must be **.END**. Do not leave a line after the **.END** command.

The output file for this example looks like:

08/13/02 SPICE 2g.6 08:09:56

Test Program for Diode Circuit in Problem Fig. 3

Input Listing Temperature = 27.000 Deg C

```
*This is a comment. In some SPICE outputs E=D for 10
.model diodel d is=6e-9 n=1.9
d1 3 0 diodel
R1 2 3 500
V1 1 0 dc 6
Vs 2 1 ac 0.001
.dc V1 6 6.1 0.02
.tf v(3) V1
.ac lin 3 1.66 3.66
.OP
.print dc v(3,0) i(V1) v(1,0)
.print ac v(3,0) i(vs)
.END
```

08/13/02 SPICE 2g.6 08:09:56

Test Program for Diode Circuit in Problem Fig. 3

Diode Model Parameters Temperature = 27.000 Deg C

```
diodel
is 6.00D-09
n 1.900
```

08/13/02 SPICE 2g.6 08:09:56

Test Program for Diode Circuit in Problem Fig. 3

DC Transfer Curves Temperature = 27.000 Deg C

v1	v(3)	i(v1)	v(1)
6.000E+00	7.068E-01	-1.059E-02	6.000E+00
6.020E+00	7.070E-01	-1.063E-02	6.020E+00
6.040E+00	7.072E-01	-1.067E-02	6.040E+00
6.060E+00	7.074E-01	-1.071E-02	6.060E+00
6.080E+00	7.076E-01	-1.074E-02	6.080E+00
6.100E+00	7.077E-01	-1.078E-02	6.100E+00

y 08/13/02 SPICE 2g.6 08:09:56

Test Program for Diode Circuit in Problem Fig. 3

Small Signal Bias Solution Temperature = 27.000 Deg C

node voltage node voltage node voltage

(1) 6.0000 (2) 6.0000 (3) 0.7068

Voltage source currents

Name current

v1 -1.059D-02

vs -1.059D-02

Total power dissipation 6.35D-02 Watts

08/13/02

SPICE 2g.6

08:09:56

Test Program for Diode Circuit in Problem Fig. 3

Operating Point Information Temperature = 27.000 Deg C

**** Diodes

d1
 model diode1
 id 1.06E-02
 vd 0.707
 req 4.64E+00
 cap 0.00E+00

DC Operating Point of D1

Small signal model parameters

**** Small-Signal Characteristics

v(3)/v1 = 9.199D-03
 input resistance at v1 = 5.046D+02
 output resistance at v(3) = 4.599D+00

Small signal Increment .TF

08/13/02

SPICE 2g.6

08:09:56

Test Program for Diode Circuit in Problem Fig. 3

AC Analysis

Temperature = 27.000 Deg C

freq	v(3)	i(vs)
1.660E+00	9.199E-06	1.982E-06
2.660E+00	9.199E-06	1.982E-06
3.660E+00	9.199E-06	1.982E-06

y Job concluded

Total job time 0.02